“Creatures” Tutorial – Part II

OK, OK, I’m finally getting around to this!

Part I of the tutorial left off with a pretty lifeless creature in my own estimation. Yes, it had its parts, and they were connected, and reacted to each other and everything, but unless you throw it around, it just kind of sinks to the bottom and lies there, like something that fell in the lake and drowned. Of course, I was banking on the idea that you would take the code, and stuff you learned and expand on it. Several people did just that and sent me some pretty cool swf’s. But, now let’s take it to the next level.

I’m going to start this pretty much from scratch, so it may or may not flow smoothly from the last tutorial, but if you understood everything there, you should have no problem with this one. Naturally, I am learning all the time, too. So this file may have some optimizations I hadn’t even thought of in the part one.

Today, kids, we’re going to make this creature walk.

1. First, create some body parts. You’re going to need a body and a couple of feet. Make it simple for now. When you get it working, you can go back and get artistic with it. Here’s mine:


The body is 30 pixels across, the feet are 10. Name the body instance “body” and the feet “f0” and “f1”.

2. In the original file, the feet and legs could be at any angle in relation to the body. This accounted for its rag doll appearance and behavior. In this file, we will give our legs some bones, in other words, an ideal angle that they should be at. Of course, different forces can act on them and push them one way or the other, but they will always attempt to spring back to their ideal angle.

In visual terms, one leg will stick off the right side of the body, and one off the left side of the body. In clock terms, one at 3:00, one at 9:00. In degrees, 0 and 180. In radians (getting to the point here), 0 and Math.PI. So we’ll code that right into each foot as our first code in frame one, on the main time line:

f0.angle = 0;

f1.angle = Math.PI;

3. Next, we need to determine some other constants. The variable, “leg” will be the ideal distance between foot and body. Then we have gravity, k and damping, which you should know from Part I.

leg = 50;

k = .2;

damp = .9;

grav = .5;

4. Now we are ready to bring our feet to life by giving them some onEnterFrame code to run. Well, there won’t be much life just yet, but it will start to hold together.

f0.onEnterFrame = footMove;

f1.onEnterFrame = footMove;

function footMove(){


var tx = body._x – Math.cos(this.angle)*leg;


var ty = body._y – Math.sin(this.angle)*leg;


var ax = (tx – this._x)*k;


var ay = (ty – this._y)*k;


this.vx += ax;


this.vy += ay;


this.vx *= damp;


this.vy *= damp;


this._x += this.vx;


this._y += this.vy;

}


Let’s go through the footMove function step by step, to make sure we are on the same page.

First, we get a target x and y (tx, ty). This will be 75 pixels (value of leg) away from the body, at the angle specified by the foot’s angle property. So foot 0 will have a target 75 pixels to the right of the body, and foot 1’s target is 75 pixels to the left. Then we perform a simple spring action, setting a x and y acceleration factor which is distance to the target times k. Add acceleration to velocity, damp it, and add velocity to position. Voila! The feet now spring to opposite sides of the body.

5. Next, we can add some lines to represent the legs. I prefer to do this in a separate onEnterFrame function.

onEnterFrame = function(){


clear();


lineStyle(1,0,100);


moveTo(f0._x, f0._y);


lineTo(body._x, body._y);


lineTo(f1._x, f1._y);

}

This simply clears the screen and draws a line from foot 0 to body to foot 1.

6. So far, so good. Now most animals on this planet have their legs and feet below their body, not sticking off in the air. That’s pretty simple. We’ll just move the targets down a bit. Here’s the whole thing so far, with the changes in bold:

f0.angle = 0;

f1.angle = Math.PI;

leg = 50;

k = .2;

damp = .9;

grav = .5;

f0.onEnterFrame = footMove;

f1.onEnterFrame = footMove;

function footMove(){


var tx = body._x – Math.cos(this.angle)*leg;


var ty = body._y – Math.sin(this.angle)*leg+50;


var ax += (tx – this._x)*k;


var ay += (ty – this._y)*k;


this.vx += ax;


this.vy += ay;


this.vx *= damp;


this.vy *= damp;


this._x += this.vx;


this._y += this.vy;

}
onEnterFrame = function(){


clear();


lineStyle(1,0,100);


moveTo(f0._x, f0._y);


lineTo(body._x, body._y);


lineTo(f1._x, f1._y);

}

All we did here is add 50 to the ty, the y target for each foot. This makes it 50 pixels lower than the body. Now it looks like something that might actually be able to walk.

7. Before we actually set up the walking action and applying gravity, I wanted to enhance the legs a little bit. Might as well have a creature with nice legs, right? At this point, they stick out like two twigs. But we can use the power of the drawing API to make them a bit more realistic. We’ll change the lineTo’s to curveTo’s:

onEnterFrame = function(){


clear();


lineStyle(1,0,100);


moveTo(f0._x, f0._y);


curveTo(f0._x, f0._y-50, body._x, body._y);


curveTo(f1._x, f1._y-50, f1._x, f1._y);

}


This just changes the lineTo’s to curveTo’s and uses a control point of the foot’s _x and its _y minus 50. You can adjust that 50 to get a different shaped leg. Now that it looks good, let’s get it moving.

8. Making the creature walk is as simple as changing the angles that the feet use to calculate their target points. Rather than actually change each foot’s angle property, we’ll create another, general angle variable and change that. Then, the feet will add the overall angle to their own personal angle to find their target. We’ll store this angle right in the body movie clip. Note, that this angle won’t actually affect the body at all, but will hold the value so that the feet know where to find it. Here’s the code so far, with these changes:

f0.angle = 0;

f1.angle = Math.PI;

body.angle = 0;

leg = 50;

k = .2;

damp = .9;

grav = .5;

f0.onEnterFrame = footMove;

f1.onEnterFrame = footMove;

function footMove(){


var realAngle = this.angle + body.angle;


var tx = body._x – Math.cos(realAngle)*leg;


var ty = body._y – Math.sin(realAngle)*leg+50;


var ax = (tx – this._x)*k;


var ay = (ty – this._y)*k;


this.vx += ax;


this.vy += ay;


this.vx *= damp;


this.vy *= damp;


this._x += this.vx;


this._y += this.vy;

}
onEnterFrame = function(){


clear();


lineStyle(1,0,100);


moveTo(f0._x, f0._y);


curveTo(f0._x, f0._y-50, body._x, body._y);


curveTo(f1._x, f1._y-50, f1._x, f1._y);

}

First, we set body.angle to zero. Then in the footMove we create a new variable, realAngle. This is the sum of the foot’s angle plus body.angle. We use realAngle to determine our targets.

Of course, since body.angle is just sitting at zero, you don’t see any action or even any difference from the last version. So, let’s get moving.

9. We’ll change body.angle with an onEnterFrame handler for body. For now, we’ll simply increase it a bit. Remember that the angles here are in radians, so you only have to increase them by a fraction to see a difference:

body.onEnterFrame = function(){


this.angle += .05;

}
Yay! Now we have something that actually looks like it’s walking!

10. Now, one important point here is that so far there is only a one way connection between body and feet. The feet spring away from the body, but the body is pretty static. I discovered a pretty simple way to figure out the forces acting on the body. We’ve already gone through all the trouble of calculating targets and the resulting forces for each foot. The forces on the body are exactly the opposite. The foot springs toward the body, the body springs toward the foot. These forces are ax and ay in the footMove function. We are adding them to the foot’s velocity in the following lines:


this.vx += ax;


this.vy += ay;

If we want to apply the opposite force to the body…we simply add a negative ax and ay to the body’s velocity! Or, in simpler terms, we subtract ax and ay from body’s velocity:


this.vx += ax;


this.vy += ay;


body.vx -= ax;


body.vy -= ay;

Now, we just need to fill in the body.onEnterFrame function to react to its newfound velocities.

body.onEnterFrame = function(){


this.angle += .05;


this.vx *= damp;


this.vy *= damp;


this._x += this.vx;


this._y += this.vy;

}
And here’s the file as it stands, with all the recent changes.

f0.angle = 0;

f1.angle = Math.PI;

body.angle = 0;

leg = 50;

k = .2;

damp = .9;

grav = .5;

f0.onEnterFrame = footMove;

f1.onEnterFrame = footMove;

function footMove(){


var realAngle = this.angle + body.angle;


var tx = body._x – Math.cos(realAngle)*leg;


var ty = body._y – Math.sin(realAngle)*leg+50;


var ax = (tx – this._x)*k;


var ay = (ty – this._y)*k;


this.vx += ax;


this.vy += ay;


body.vx -= ax;


body.vy -= ay;


this.vx *= damp;


this.vy *= damp;


this._x += this.vx;


this._y += this.vy;

}
onEnterFrame = function(){


clear();


lineStyle(1,0,100);


moveTo(f0._x, f0._y);


curveTo(f0._x, f0._y-50, body._x, body._y);


curveTo(f1._x, f1._y-50, f1._x, f1._y);

}

body.onEnterFrame = function(){


this.angle += .05;


this.vx *= damp;


this.vy *= damp;


this._x += this.vx;


this._y += this.vy;

}
(Also, on the stage, you might want to initially place the feet in the approximate positions relative to the body that they will actually spring to. If they are very far away, they will spring with a huge force right away, possibly blowing the creature right off stage.)

11. OK, now we just need to throw in some physics and an external environment. We’ve already defined gravity, so it’s easy enough to add that to the body and feet just before damping is applied, with the line:

this.vy += grav;

But once we apply gravity, we’ll need some ground for the creature to land on. For the sake of simplicity, let’s assume that the movie is 400 pixels high and hard code 400 as the ground point. I chose to only have the feet to react to the ground. Since the body always tries to be above the feet, it will rarely hit bottom. Here’s the whole thing so far, again with changes in bold:

f0.angle = 0;

f1.angle = Math.PI;

body.angle = 0;

leg = 50;

k = .1;

damp = .9;

grav = .5;

f0.onEnterFrame = footMove;

f1.onEnterFrame = footMove;

function footMove(){


var realAngle = this.angle + body.angle;


var tx = body._x – Math.cos(realAngle)*leg;


var ty = body._y – Math.sin(realAngle)*leg+50;


var ax = (tx – this._x)*k;


var ay = (ty – this._y)*k;


this.vx += ax;


this.vy += ay;


this.vy += grav;


body.vx -= ax;


body.vy -= ay;


this.vx *= damp;


this.vy *= damp;


this._x += this.vx;


this._y += this.vy;


if(this._y>400){



this._y = 400;



this.vx = 0;



this.vy = 0;


}

}
onEnterFrame = function(){


clear();


lineStyle(1,0,100);


moveTo(f0._x, f0._y);


curveTo(f0._x, f0._y-50, body._x, body._y);


curveTo(f1._x, f1._y-50, f1._x, f1._y);

};

body.onEnterFrame = function(){


this.angle += .05;


this.vy += grav;


this.vx *= damp;


this.vy *= damp;


this._x += this.vx;


this._y += this.vy;

};
A couple things to comment on here. One is that the _y measurement of the foot will be its registration point. So, you want that to be the lowest point of the foot, as shown in the following diagram:


This will ensure that the bottom of the foot hits the ground exactly, and doesn’t sink into it, or ride above it.

Next we have the lines where the foot hits:


if(this._y>400){



this._y = 400;



this.vx = 0;



this.vy = 0;


}

First, we firmly place our feet on the floor by setting _y to 400. Then, we set its vx and vy to zero. This effectively stops the foot from moving. If you’ve done the gravity tutorial, or the previous creature tutorial, you are probably used to setting vy to –vy by saying this.vy *= -1 or something of the sort. In this case, we don’t want it to bounce, we just want it to stop. But, by all means experiment with some bouncy feet. It can get interesting.

Also, in previous files, we wouldn’t change vx at all. We need to do that here to get some traction on the ground. If you remove that line, your creature will look like he is running on ice. The foot will stop it’s downward motion, but will keep slipping in the same horizontal direction. Although we stop the x motion of the foot, the opposite x motion of the body remains, which drives it forward. Thus the body actually does perform a walking action.

Conclusion
Well, that’s all for today’s segment! Now, I realize that there will definitely need to be a Part III (at least). Again, I’ve left you with a skeleton of a file with loads of room for experimentation. I’m going to give you a couple of suggestions on directions to take it from here:

1. Add more feet! You just need to give each one an angle between 0 and Math.PI*2, and make sure you assign footMove as its onEnterFrame handler. Oh, and don’t forget to draw lines to them so they aren’t just hanging in space.


2. Make it walk faster or slower, and the opposite direction. This is done by changing the line:



this.angle += .05;

in body.onEnterFrame. I suggest you change this to:

this.angle += speed;

You can then change the value of speed based on keyboard input or mouse position or whatever creative ideas you might have for it. Remember, a higher speed moves faster to the right. A negative speed will move it to the left.

3. Make some walls for it to bump into. 


4. Make another creature and have them interact.

As always, if you come up with anything cool, I would love to see it. 

Keith Peters

kp@bit-101.com
Have you had your BIT-101 today?

